

MAX3271E/D Output Model

SPICE I/O Macromodels aid in understanding signal integrity issues in electronic systems. Most of Maxim's High Frequency/Fiber Communication ICs utilize input and output (I/O) circuits with Current Mode Logic (CML), Positive Emitter Coupled Logic (PECL), and Low Voltage Differential Signal (LVDS) formats to transfer data into and out of an IC. These models are based on simplified circuit expressions that may include replacement of active circuit elements with ideal controlled voltage and current sources. As such, simulation with macromodels should be treated as 'typical' performance and not relied upon as final proof-of-design. Use of macromodel descriptions is not a substitute for worst-case design analysis, nor for testing real circuits over temperature, supply, and other operating limits.

The output format is provided as ASCII text netlists suitable for generic SPICE. This format is compatible with most versions of SPICE such as PSPICE and HSPICE. Additional information is found in HFAN 6.1 *Input/Output Models for Maxim Fiber Components*.

To extract the circuit netlists using the Adobe Acrobat Reader follow these instructions. Select the "Text Select Tool" by clicking the left mouse button on this icon of the menu bar (a capital T with a small dashed box to the lower right). Highlight the desired netlist text with the cursor. Use the copy command from the edit menu to capture the selected lines. Then paste the selected lines into a text file editor and save the file with an extension compatible with the simulator.

Revision A0, August 3, 2004

Output Model for the MAX3271




Figure 1. Output model for signal OUT of the MAX3271.

Notes:

The schematics on the previous page represent the output stage of the Maxim MAX3271 2.5Gbps Transimpedance Preamplifier. The output circuit shown is for the signal outputs (OUT+, OUT-). However the models are given in generic SPICE, which only accepts node names as numbers. As discussed in the application note, the output signals are described as (2001, 2002). This model is only valid at 25°C. The bias current for the output circuitry is modeled by an ideal current source. This model is not compensated for variations in VCC, so VCC equal to 3.3V.

The Output Stage: The output stage of the MAX3271 is shown as the sub-circuit “OUTDRV”.

The OUTDRV Sub-circuit: The outdrv sub-circuit is a simplified version of the output stage used by the MAX3271 Transimpedance Preamplifier. The output stage is driven by a differential amplifier and terminated differentially with 100Ω. The output waveform is configured to be at a differential voltage of 300mV peak to peak. The output waveform has a frequency of 1.25Ghz with rise and fall times of approximately 90ps. In Appendix A, the netlist is given in SPICE 2G6 format.

Text File Format: This model is shipped in “pdf” format. Models and netlists can be copied to text format in the Acrobat Reader by holding the left mouse button on the “Text Select Tool.” Then the user can “select” what netlist and/or subcircuit with the mouse. Then use the copy command from the “edit” menu to capture the selected lines. These lines can then be “pasted” into the user’s text file.

Appendix A: Output Netlist

* 3271 Output Model

.OPT ACCT NOMOD LIMPTS=10000

.TEMP 25

.OP

.TRAN 2P 2n

* Voltage Source

VCC 101 0 3.3

* Load Resistance

RL 2001 2002 100

XOUTDRV 2001 2002 101 OUTDRV

.SUBCKT OUTDRV 2001 2002 101

VIN1 1 0 PULSE (1 1.5 0.04n 50p 50p 350p 800p)

VIN2 3 0 PULSE (1.5 1 0.04n 50p 50p 350p 800p)

R1 2 1 50

R2 4 3 50

R3 101 2001 50

R4 101 2002 50

C1 2 0 .2p

C2 4 0 .2p

C3 2001 2002 103.1f

C4 2002 2001 103.1f

C5 2001 2002 125.4f

C6 2002 2001 125.4f

C7 2001 0 1.746p

C8 2002 0 1.746p

XQ1 2001 2 5 0 N102M066_10

XQ2 2002 4 5 0 N102M066_10

I1 5 0 6.06mA

.ENDS OUTDRV

```
.SUBCKT N102M066_10 1 2 3 21
CP1SUB 2 201 15.642F
RP1SUB 20 201 100K
CTRENCH 1 202 78.812F
RFIELDDEPI 202 21 121.753
RREVERT 202 20 1G
CBL 10 20 19.782F
RSUB 20 21 33.861K
CWAFFER 20 21 15.558F
CP1EPI 10 12 19.174F
CP1P2 12 3 22.114F
RBX 2 12 7.187 TC=1.934M
RCX 1 10 2.032 TC=2.640M,410.600N
RCI 10 11 507.968M TC=2.640M,410.600N
REX 13 3 1.270
QN 11 12 13 11 TX 10
.MODEL TX NPN( IS=6.926E-018 XTI=3 EG=1.120 BF=380 BR=12 XTB=0 VAF=66
+ VAR=2.500 NF=1.018 NR=1.020 NE=2 NC=1.560 IKF=15.238M IKR=432.900U
+ ISE=3.463E-018 ISC=1.558E-030 RB=17.968 RBM=13.476 IRB=1.558M
+ CJE=16.207F MJE=463M VJE=1.100 FC=990M CJC=7.862F MJC=350M VJC=1
+ TF=1.320P TR=5N XTF=2 VTF=1.200 ITF=56.277M PTF=5 KF=12.979N
+ AF=2.150 )
.ENDS N102M066_10
```

```
.SUBCKT PADES50 2 3 4 5
XP1 2 5 PAD4SQ3P7
XQ1 2 3 5 DE0396
XQ2 4 2 5 DE0396
.ENDS PADES50
```

```
.SUBCKT DE0396 1 2 21
CTRENCH 2 202 33.018F
RFIELDDEPI 202 21 291.149
RREVERT 202 20 1G
CBL 4 5 37.842F
RSUB 5 21 49.887K
CWAFFER 5 21 10.560F
CP1EPI 1 4 32.900F
DD 1 4 DCB
RS 4 2 12.010 TC=4.306M,4.262U
```

.MODEL DCB D(IS=3.485E-018 N=1.050 CJO=95.040F VJ=800M M=500M)
.ENDS DE0396

.SUBCKT PAD4SQ3P7 1 3
CPAD 1 10 67.534F
REPI 10 20 378.507 TC=4.800M,5U
CTRENCH 21 20 22.531F
CBL 21 20 1.413P
RX 20 21 1G
RS 3 21 5.416K
CWAFER 21 3 2.587F
.ENDS PAD4SQ3P7

.PROBE
.END